
XSA Parallel Port Interface 

November 16, 2001 (Version 1.0) Application Note by D. Vanden Bout 

Summary 

This application note describes the default parallel port interface circuit that is programmed into the XC9572XL 
CPLD on the XSA Board.  It also discusses how to change the parallel port interface to support other features. 

 
The Default Parallel Port Interface 

Listing 1 shows the VHDL code for the default 
parallel port interface that is programmed into the 
XC9572XL CPLD on the XSA Board.  This interface 
provides two functions: 

• It transfers configuration bitstreams from the PC 
to the SpartanII FPGA. 

• It lets the PC and the SpartanII communicate 
through the parallel port after the FPGA is 
configured. 

How the VHDL implements these functions is 
described below. 

Lines 39–42 disable other chips and functions on the 
XSA Board so they cannot interfere with the 
configuration of the SpartanII device.  The JTAG 
circuitry of the SpartanII FPGA is kept quiescent by 
holding its clock pin low on line 41.  The Flash RAM 
is disabled by pulling its chip-enable pin high on line 
42.  The Flash chip-enable is allowed to float once 
the SpartanII device is configured (as indicated when 
the SpartanII DONE pin goes high) so the SpartanII 
can take control of the Flash chip-enable. 

The circuitry that actually controls the configuration of 
the SpartanII device is described on lines 44–60.  
The CPLD pulls down the M0 mode pin of the 
SpartanII device to set the FPGA into the Slave 
Parallel configuration mode (the M1 and M2 mode 
pins of the SpartanII are hard-wired on the XSA 
PCB).  The PROGRAM pin for the SpartanII is 
connected to data line D7 of the parallel port on line 
46.  A low level on D7 will initiate the configuration of 
the SpartanII and its DONE pin will go low.  The low 
level on the DONE pin forces low levels on the chip-
select and write-strobe of the SpartanII on lines 47-
48.  This enables the writing of byte-wide 
configuration data into the SpartanII.  These pins are 
released after DONE goes high because they 

become general-purpose I/O pins after configuration 
is completed. 

The configuration data bytes arrive as two four-bit 
nybbles over data lines D2–D5.  The upper nybble of 
each configuration byte is stored in the config_data 
register on the rising edge of the cclk (lines 55–60).  
The inverse of parallel port data line D0 drives the 
internal cclk signal, so the data is latched into the 
config_data register on the falling edge of D0. 

The upper nybble in the config_data register is 
concatenated with the lower nybble on the D2–D5 
data lines to form a complete byte of configuration 
data.  This configuration data is passed to the 
SpartanII on line 51.  The configuration clock for the 
SpartanII is the inverse of the clock that controls the 
config_data register (line 50).  So the configuration 
data is latched into the SpartanII on the falling edge 
of cclk.  The overall process of getting byte n of 
configuration data into the SpartanII looks like this: 

cclk

ppd(5..2)

ppd0 and
S2_cclk

n(7..4)

n(7..4)

n+1(7..4)

n+1(7..4)

n(3..0)

config_data

upper nybble latched

configuration byte
latched into SpartanII

 

Once all the configuration data enters the SpartanII, it 
will raise its DONE pin.  This will raise the chip-select 
and write-strobe of the SpartanII (lines 47–48) and 
disable further writing of configuration data.  It will 
also make the CPLD release control of the SpartanII 
configuration data input pins (lines 52 and 53).  

These pins become general purpose I/O pins after 
DONE goes high and the FPGA can use them to 
access data from the Flash or to drive the seven-
segment LED.  If the SpartanII is not accessing the 
Flash (i.e., the Flash chip-enable, fceb, is high) then 
the CPLD will light up the decimal-point segment of 
the LED to indicate that the SpartanII is currently 
configured while releasing the remaining data lines 
(line 52).  But if the SpartanII enables the Flash, then 
the CPLD releases all the data lines to prevent 
contention with the FPGA (line 53). 

Lines 63 and 64 describe how the CPLD connects 
the parallel port data lines to the general-purpose I/O 
pins of the SpartanII.  Line 63 connects the eight 
parallel port data lines to the SpartanII so it can 
receive data from the PC.  (The lower two data bits 
are inverted to mimic the parallel port connection 
used by the XS95, XS40 and XSV Boards.)  These 
same FPGA pins are used by the SpartanII to 
connect to address pins A8–A13, the write-enable 
and the output-enable of the Flash.  Therefore, the 
CPLD releases these pins to avoid contention when 
the SpartanII lowers the Flash chip-enable and 
begins to access the Flash. 

On line 64 the CPLD connects three of the parallel 
port status lines to the SpartanII so the FPGA can 
transfer data back to the PC.  The SpartanII also has 
a direct connection to a fourth status line.  (The 
CPLD uses the fifth status line for the TDO output of 
its JTAG interface, so it is not available for general-
purpose transfer of data to the PC.) 

Finally, the CPLD connects the output of the 
programmable oscillator on the XSA Board to a 
dedicated clock input of the SpartanII (line 67). 

The default parallel port interface implements the 
following connections between the SpartanII and the 
parallel port, LED and programmable oscillator. 

 

 

 

 

 

 

 

 

 

Pin Function SpartanII Pin# 

Parallel Port Data Pin D0 50 

Parallel Port Data Pin D1 48 

Parallel Port Data Pin D2 42 

Parallel Port Data Pin D3 47 

Parallel Port Data Pin D4 65 

Parallel Port Data Pin D5 51 

Parallel Port Data Pin D6 58 

Parallel Port Data Pin D7 43 

Parallel Port Status Pin S3 40 

Parallel Port Status Pin S4 29 

Parallel Port Status Pin S5 28 

Parallel Port Status Pin S6 78 

LED Segment S0 67 

LED Segment S1 39 

LED Segment S2 62 

LED Segment S3 60 

LED Segment S4 46 

LED Segment S5 57 

LED Segment S6 49 

LED Segment DP 44 

Programmable Oscillator 88 

Changing the Parallel Port Interface 

The parallel port interface is stored in the nonvolatile 
Flash of the XC9572XL CPLD on the XSA Board.  
Any design you load into the CPLD will become 
active as soon as the XSA Board powers up.  So it is 
possible to load a faulty interface design into the 
CPLD that makes it impossible to program the 
SpartanII even after you cycle the power.  The only 
solution is to explicitly reprogram the CPLD with a 
functional interface using GXSLOAD.  Then the XSA 
Board will function correctly again. 

When generating a new interface for the CPLD, you 
must set the USERCODE signature register to the 
four-character string <4>!.  The XSTOOLs utilities 
look for this signature in the CPLD to verify that a 
valid interface is present. 

There are probably only three things you might want 
to change in the default parallel port interface: 

1. You might want to give the SpartanII access 
to the decimal-point of the seven-segment 
LED instead of having the CPLD use it to 
display the configuration status of the FPGA.  
To do this, change the bitstring on line 52 to 
“ZZZZZZZZ”.  Note that once you do this you 
will no longer get any visual indication of 

whether your SpartanII bitstreams were 
downloaded correctly into the FPGA. 

2. You might want to use all eight parallel port 
data pins to send data to the SpartanII.  
Therefore you will have to prevent any data 
presented on bit D7 from affecting the 
PROGRAM pin of the SpartanII or it will 
erase its configuration.  As an alternative, 
you could change line 46 to connect the 
PROGRAM pin to the active-low pushbutton 
on the XSA Board so that you can start a 
reconfiguration of the FPGA by pressing this 
button.  Or you could connect the 
PROGRAM pin to a constant logic ‘1’ and 
then you would have to interrupt the power to 
the XSA Board to force a reconfiguration. 

3. You might want to change the clock that is 
passed to the SpartanII on line 67.  You can 
pass the programmable oscillator output 
through a divider and on to the SpartanII.  Or 
you could drive the SpartanII clock input with 
one of the parallel port data pins to do single-
stepping of a design.  Or you could have the 
CPLD release the SpartanII clock input 
entirely so you can drive it with an external 
clock source. 

Listing 1: VHDL code for the default CPLD parallel port interface. 
library ieee; 1 
use ieee.std_logic_1164.all; 2 
 3 
entity dwnldpar is 4 
 port( 5 
  -- parallel port data and status pins 6 
  ppd: in std_logic_vector(7 downto 0); 7 
  pps: out std_logic_vector(5 downto 3); 8 
 9 
  -- programmable oscillator 10 
  clk: in std_logic; 11 
   12 
  -- Spartan2 FPGA pins 13 
  S2_tck: out std_logic; -- driver to Spartan2 JTAG clock 14 
  S2_cclk: out std_logic; -- driver to Spartan2 config clock 15 
  S2_progb: out std_logic; -- driver to Spartan2 program pin 16 
  S2_csb: out std_logic; -- driver to Spartan2 config. chip-select 17 
  S2_wrb: out std_logic; -- driver to Spartan2 config. write strobe 18 
  S2_initb: in std_logic; -- input from Spartan2 init pin 19 
  S2_done: in std_logic; -- input from Spartan2 done pin 20 
  S2_d: out std_logic_vector(7 downto 0); -- drivers to Spartan2 data pins and 7-seg LED 21 
  S2_m: out std_logic_vector(0 downto 0); -- Spartan2 config mode pins 22 
  S2_clk: out std_logic; -- clock output to Spartan2 23 
  S2_ppd: out std_logic_vector(7 downto 0); -- parallel port data pins to Spartan2 24 
  S2_pps: in std_logic_vector(5 downto 3);-- parallel port status pins from Spartan2 25 
  fceb: inout std_logic -- Flash chip-enable 26 
 ); 27 
end entity dwnldpar; 28 
 29 
 30 
architecture arch of dwnldpar is 31 
 constant LO: std_logic := '0'; 32 
 constant HI: std_logic := '1'; 33 
 constant SLAVE_PARALLEL_MODE: std_logic_vector(0 downto 0) := "0"; 34 
 signal cclk: std_logic; 35 
 signal config_data: std_logic_vector(3 downto 0); 36 
  37 
 begin 38 
 -- disable other chips on the XSV Board so they don't interfere 39 
 -- during the configuration of the Spartan2 FPGA 40 
 S2_tck  <= LO; -- deactivate Spartan2 JTAG circuit 41 
 fceb  <= HI when S2_done=LO else 'Z'; -- disable Flash during config. 42 
 43 
 -- connect Spartan2 configuration pins 44 
 S2_m  <= SLAVE_PARALLEL_MODE; -- set Spartan2 config mode pins 45 
 S2_progb <= ppd(7);-- Spartan2 programming pulse comes from parallel port 46 
 S2_csb  <= LO when S2_done=LO else 'Z'; -- enable writing of data 47 
 S2_wrb  <= LO when S2_done=LO else 'Z'; -- during Spartan2 config. phase 48 
 cclk   <= not ppd(0); 49 
 S2_cclk  <= not cclk; 50 
 S2_d   <= (config_data & ppd(5 downto 2)) when S2_done=LO else 51 
     "ZZZZZZ1Z" when fceb=HI else  -- show Spartan2 config status on LED-DP 52 
     "ZZZZZZZZ"; -- release control of Spartan2 Flash/LED pins 53 
 54 
 process(cclk) 55 
 begin 56 
  if(cclk'event and cclk=HI) then 57 
   config_data <= ppd(5 downto 2); 58 
  end if; 59 
 end process; 60 
  61 
 -- connect the parallel port data and status pins to the Spartan2 62 

 S2_ppd  <= (ppd(7 downto 2) & not(ppd(1 downto 0))) when fceb=HI else "ZZZZZZZZ"; 63 
 pps   <= S2_pps; 64 
  65 
 -- send the programmable oscillator clock to the Spartan2 66 
 S2_clk  <= clk; 67 
  68 
end architecture arch; 69 
 70 

Listing 2: User-constraint file for CPLD pin assignments. 
# 1 
# pin assignments for the XC9572XL CPLD chip on the XSA Board 2 
# 3 
 4 
# Spartan2 FPGA connections to CPLD 5 
net S2_clk  loc=p42; 6 
net S2_tck  loc=p13; 7 
# net S2_dout  loc=p18; 8 
# net S2_din  loc=p2; 9 
net S2_wrb  loc=p19; 10 
net S2_csb  loc=p15; 11 
# net S2_initb loc=p38; 12 
net S2_done  loc=p40; 13 
net S2_progb  loc=p39; 14 
net S2_cclk  loc=p16; 15 
net S2_m<0>   loc=p36; 16 
net S2_d<0>  loc=p2; 17 
net S2_d<1>  loc=p4; 18 
net S2_d<2>  loc=p5; 19 
net S2_d<3>  loc=p6; 20 
net S2_d<4>  loc=p7; 21 
net S2_d<5>  loc=p8; 22 
net S2_d<6>  loc=p9; 23 
net S2_d<7>  loc=p10; 24 
net S2_ppd<0>  loc=p45; 25 
net S2_ppd<1>  loc=p44; 26 
net S2_ppd<2>  loc=p57; 27 
net S2_ppd<3>  loc=p43; 28 
net S2_ppd<4>  loc=p56; 29 
net S2_ppd<5>  loc=p46; 30 
net S2_ppd<6>  loc=p49; 31 
net S2_ppd<7>  loc=p12; 32 
net S2_pps<3>  loc=p1; 33 
net S2_pps<4>  loc=p64; 34 
net S2_pps<5>  loc=p63; 35 
 36 
# Flash RAM 37 
# net fd<0>  loc=p2; 38 
# net fd<1>  loc=p4; 39 
# net fd<2>  loc=p5; 40 
# net fd<3>  loc=p6; 41 
# net fd<4>  loc=p7; 42 
# net fd<5>  loc=p8; 43 
# net fd<6>  loc=p9; 44 
# net fd<7>  loc=p10; 45 
# net fa<0>  loc=p1; 46 
# net fa<1>  loc=p64; 47 
# net fa<2>  loc=p63; 48 
# net fa<3>  loc=p62; 49 
# net fa<4>  loc=p61; 50 
# net fa<5>  loc=p60; 51 
# net fa<6>  loc=p59; 52 
# net fa<7>  loc=p58; 53 
# net fa<8>  loc=p45; 54 
# net fa<9>  loc=p44; 55 
# net fa<10>  loc=p57; 56 
# net fa<11>  loc=p43; 57 
# net fa<12>  loc=p56; 58 
# net fa<13>  loc=p46; 59 
# net fa<14>  loc=p47; 60 
# net fa<15>  loc=p52; 61 
# net fa<16>  loc=p51; 62 

# net fa<17>  loc=p48; 63 
# net frstb  loc=p50; # Flash reset 64 
# net foeb  loc=p12; # Flash output-enable 65 
# net fweb  loc=p49; # Flash write-enable 66 
net fceb   loc=p11; # Flash chip-enable 67 
 68 
# DIP and pushbutton switches 69 
# net dipsw<1> loc=p47; 70 
# net dipsw<2> loc=p52; 71 
# net dipsw<3> loc=p51; 72 
# net dipsw<4> loc=p48; 73 
 74 
# 7-segment LEDs 75 
# net s<0>  loc=p10; 76 
# net s<1>  loc=p2; 77 
# net s<2>  loc=p9; 78 
# net s<3>  loc=p8; 79 
# net s<4>  loc=p5; 80 
# net s<5>  loc=p7; 81 
# net s<6>  loc=p6; 82 
# net dp   loc=p4; 83 
 84 
# programmable oscillator 85 
net clk   loc=p17; 86 
 87 
# parallel port 88 
net ppd<0>     loc=p33; 89 
net ppd<1>     loc=p32; 90 
net ppd<2>  loc=p31; 91 
net ppd<3>  loc=p27; 92 
net ppd<4>  loc=p25; 93 
net ppd<5>  loc=p24; 94 
net ppd<6>  loc=p23; 95 
net ppd<7>  loc=p22; 96 
net pps<3>  loc=p34; 97 
net pps<4>  loc=p20; 98 
net pps<5>  loc=p35; 99 

